Q:

Skip first worksheet of the workbook in VBA

I want to have a number of worksheet on each of worksheets in a workbook skipping the first one and do some formatting as well, however i want this vba code to skip the first worksheet (name can differ but always is going to be first). Thus the question is how should i do that?
Sub ex2()
    Dim kl As Worksheet
    Dim Ws_Count As Integer
    Dim a As Integer
    Ws_Count = ActiveWorkbook.Worksheets.Count
    For a = 2 To Ws_Count

        With Rows("2:2")
            .RowHeight = 20
            .Interior.Color = RGB(150, 250, 230)
        End With
        With Range("B2")
            .Value = "Sheet Number" & " " & a
            .Font.Size = 12
            .Font.Bold = True
            .Font.Underline = True
        End With
    Next a
End Sub

A:

Your code was good, you were only missing a single line, checking the current sheet kl.Index.
Code
Option Explicit

Sub ex2()

    Dim kl As Worksheet

    For Each kl In Worksheets
        ' not the first worksheet
        If kl.Index > 1 Then

            With kl.rows("2:2")
                .RowHeight = 20
                .Interior.Color = RGB(150, 250, 230)
            End With
            With kl.Range("B2")
                .Value = "Sheet Number" & " " & kl.Index - 1
                .Font.Size = 12
                .Font.Bold = True
                .Font.Underline = True
            End With
        End If
    Next kl

End Sub

